The Mathematica® Journal

# On Bürmann's Theorem and Its Application to Problems of Linear and Nonlinear Heat Transfer and Diffusion

## Expanding a Function in Powers of Its Derivative

**H. M. Schöpf**
**P. H. Supancic**

This article presents a compact analytic approximation to the solution of a nonlinear partial differential equation of the diffusion type by using Bürmann's theorem. Expanding an analytic function in powers of its derivative is shown to be a useful approach for solutions satisfying an integral relation, such as the error function and the heat integral for nonlinear heat transfer. Based on this approach, series expansions for solutions of nonlinear equations are constructed. The convergence of a Bürmann series can be enhanced by introducing basis functions depending on an additional parameter, which is determined by the boundary conditions. A nonlinear example, illustrating this enhancement, is embedded into a comprehensive presentation of Bürmann's theorem. Besides a recursive scheme for elementary cases, a fast algorithm for multivalued Bürmann expansions and inverse functions is developed using integer partitions. The present approach facilitates the search for expansions of analytic functions superior to commonly used Taylor series and shows how to apply these expansions to nonlinear PDEs of the diffusion type.

## Introduction

For most nonlinear problems in physics, analytic closed-form solutions are not available. Thus the investigator initially searches for an approximate analytic solution. This approximation must be reliable enough to correctly describe the dependence of the solution on all essential parameters of the system. This article aims to show that Bürmann's theorem can serve as a powerful tool for gaining approximations fulfilling such demands. We have chosen canonical examples [1, 2, 3] from the field of linear and nonlinear heat transfer to illustrate this technique.

A Bürmann series may be regarded as a generalized form of a Taylor series: instead of a series of powers of the independent variable $z - z_0$, we have a series of powers of an analytic function $\phi(z) - \phi(z_0)$:

$$f(z) = f(z_0) + \sum_{n=1}^{\infty} C_n(\phi(z) - \phi(z_0))^n.$$

Starting at an elementary level, we present a recursive calculation scheme for the coefficients of a Bürmann series. Such a recursive formulation is easily implemented in *Mathematica* and can find the Bürmann expansion for all elementary cases. For instances where we have to deal with series expansions of $f(z)$ in terms of powers of functions of the form

$$\phi(z) = \phi(z_0) + \sum_{n=\nu+1>1}^{\infty} a_n(z - z_0)^n,$$

that is, functions $\phi(z)$ for which the first $\nu$ derivatives vanish at some point $z = z_0$ of the complex plane, we approach the limits of the recursive account. To calculate such expansions using *Mathematica* efficiently, we give a generalized formulation of the coefficients of the Bürmann series, using the expansion coefficients $\mathcal{R}_k$ of the $m^{\text{th}}$ reciprocal power of an analytic function $\phi^*(z)$:

$$\frac{1}{\phi^*(z)^m} = \frac{1}{\phi^*(z_0)^m} \sum_{k=0}^{\infty} \mathcal{R}_k\,(\phi^*, m)\,(z - z_0)^k,\ \ \phi^*(z) = \frac{\phi(z) - \phi(z_0)}{z - z_0}.$$

This formulation avoids the time-consuming process of symbolic differentiation usually used. The calculation of the coefficients $\mathcal{R}_k$ is based on finding all partitions $p$ of the index $k$ in terms of the frequencies $p_s$ of the part $s$ of $p$,

$$1\,p_1 + 2\,p_2 + 3\,p_3 + \ldots + k \cdot p_k = k, \text{ with } 0 \leq p_s \leq k.$$

These sets of frequencies for the partitions are tabulated by using the function `FrobeniusSolve`. Once the coefficients $\mathcal{R}_k$ are determined by using the tabulated solutions for $p$, the calculation of the coefficients of a generalized Bürmann series is a simple task. A special case of a Bürmann series, representing a function as a series of powers of its own derivative, is of particular importance:

$$f(z) = \sum_{n=0}^{\infty} a_n(f'(z) - f'(z_0))^n.$$

Expansions of this type will be applied to functions defined by integrals. For linear and also for nonlinear processes of heat transfer, these series expansions will serve us to find valuable approximations. This is due to the fact that the integral representation for the error function leads to an expansion in fractional powers of the integrand. It turns out that a similar strategy can be applied to find approximate solutions for nonlinear cases, since these solutions obey integral equations closely related to the integral defining the error function. Finally, by introducing a free parameter, the convergence of a Bürmann expansion can be improved. The free parameter is determined by the boundary conditions. By this procedure, we find reliable analytical approximations for the heat transfer in ZnO [3], comprising only a few terms.

The common analytic solutions to these problems use Taylor series or numerical evaluations, which do not exploit the structure revealed by the integral relation fulfilled by the exact solutions. We mention here that a similar procedure can also be applied successfully to the diffusion of metal cations in a solid solvent [4].

The article is organized in such a way as to offer the formulas to the reader, together with brief remarks concerning their origin. Necessary details of deriving the formulas are displayed in the corresponding appendices.

## ■ The Elementary Approach to Bürmann's Theorem

Bürmann's theorem [5] states that it is possible to find a convergent expansion of an analytic function $f(z)$ as a sum of powers of another analytic function $\phi(z)$. The simplest form of such an expansion, supposed to be valid around some point $z = z_0$ in the complex plane, is given by

$$f(z) = f(z_0) + \sum_{n=1}^{\infty} C_n(f;\, \phi,\, z_0)\, \Phi(\phi,\, z_0,\, z)^n,\ \Phi(\phi,\, z_0,\, z) = \phi(z) - \phi(z_0), \tag{1}$$

or transferred to another notation, for some purposes more convenient,

$$f(z) = f(z_0) + \sum_{n=1}^{\infty} \mathcal{B}_n(f;\, \phi,\, z_0)\, \Psi(\phi,\, z_0,\, z)^n,\ \Psi(\phi,\, z_0,\, z) = \frac{\phi(z) - \phi(z_0)}{\phi'(z_0)}, \tag{2}$$

where the functions $\Phi$ and $\Psi$ are called the basis functions of the Bürmann series. The functions $f(z)$ and $\phi(z)$ have to fulfill certain conditions in order to guarantee the convergence of the series in (1) and (2). These conditions will be discussed later in this article. In their classic work *A Course of Modern Analysis* [5], Whittaker and Watson give a formula for the coefficient $C_n(f;\, \phi,\, z_0)$ (Bürmann coefficient) of a Bürmann series. Transferred to the notation used in (1), their formula is

$$C_n(f;\, \phi,\, z_0) = \frac{1}{n!}\, \frac{d^{n-1}}{d\zeta^{n-1}} \left(f'(\zeta)\, ((\zeta - z_0) \,/\, (\phi(\zeta) - \phi(z_0)))^n\right)_{\zeta \to z_0}. \tag{3}$$

This formula is widely cited by numerous authors [6, 7]. Actually determining the limit value of a higher-order derivative is a cumbersome procedure, which is shown in an example. Expanding the function $f(z) = z^5$ in powers of $\phi(z] = \sinh(z)$ around $z_0 = 0$ gives the following coefficients $C_i$, for which CPU time increases dramatically for $i \geqslant 10$.

```
whittakerwatsonlimit[n_, z0_] :=
 Limit[1/n! D[5 ξ^4 ((ξ - z0)/(Sinh[ξ] - Sinh[z0]))^n, {ξ, n - 1}], ξ → z0]

timecoeff = Table[Timing[whittakerwatsonlimit[i, 0]],
   {i, 10}];
Text@TableForm[timecoeff,
  TableHeadings →
   {Automatic, {"timeCPU [s]", Style["Ci", Italic]}}]
```

| | $\text{time}_{\text{CPU}}$ [s] | $C_i$ |
|---|---|---|
| 1 | 0.021622 | 0 |
| 2 | 0.109501 | 0 |
| 3 | 0.158089 | 0 |
| 4 | 0.493947 | 0 |
| 5 | 0.937498 | 1 |
| 6 | 2.103218 | 0 |
| 7 | 3.268312 | $-\frac{5}{6}$ |
| 8 | 5.041583 | 0 |
| 9 | 6.447540 | $\frac{47}{72}$ |
| 10 | 122.879467 | 0 |

## □ The Recursive Formula for Bürmann Coefficients $C_n(f; \phi, z_0)$

This section shows how to calculate the coefficients of the Bürmann series recursively, which is easier to handle than (3) and more efficient when translated to symbolic programs. If we use (1) and (2) to find convergent series representations of solutions to differential equations, it is important to simplify the algorithms necessary to determine the expansion coefficients.

For basis functions $\Phi(\phi, z_0, z)$ of the general form

$$\Phi(\phi, z_0, z) = \phi(z) - \phi(z_0) = \sum_{n=1}^{\infty} \frac{\phi^{(n)}(z_0)}{n!} (z - z_0)^n, \phi^{(1)}(z_0) = \left(\frac{d\phi}{dz}\right)_{z=z_0} \neq 0, \tag{4}$$

we get the recursion in terms of the representation used in (1),

$$C_n(f;\phi,z_0) = \frac{1}{n}\,\frac{1}{\phi'(z_0)}\,\frac{dC_{n-1}(f;\phi,z_0)}{dz_0}, \tag{5}$$

with the initial coefficient given by

$$C_1(f;\phi,z_0) = \frac{f'(z_0)}{\phi'(z_0)}. \tag{6}$$

Hence, the nested expression for $C_n(f;\phi,z_0)$ is

$$C_n(f;\phi,z_0) = \frac{1}{n!}\,\frac{1}{\phi'(z_0)}\,\frac{d}{dz_0}\left(\frac{1}{\phi'(z_0)}\,\frac{d}{dz_0}\left(\ldots\,\frac{1}{\phi'(z_0)}\,\frac{d}{dz_0}\left(\frac{f'(z_0)}{\phi'(z_0)}\right)\ldots\right)\right). \tag{7}$$

The recursion (5) is more efficient to calculate than the expression in (3) and is easily implemented.

```
c[f_, ϕ_, 1, y_] := f′[y] / ϕ′[y]

c[f_, ϕ_, n_, y_] := 1/n Together[1/ϕ′[y] D[c[f, ϕ, n - 1, y], y]]
```

The Bürmann series for $f(z)$ up to order $m$ in $\Phi(\phi,z,z_0)$ is calculated with `Bürmann[z, z0, m, f, ϕ]`.

```
Bürmann[f_, ϕ_, {z_, z0_, m_}] :=
 Module[{n, y}, f[z0] + Sum[c[f, ϕ, n, y] (ϕ[z] - ϕ[y])^n, {n, 1, m}] /. y → z0]
```

We now show the expansion for the same problem shown in the previous section (i.e. expanding $f(z) = z^5$ around $z_0 = 0$ into powers of $\phi(z) = \sinh(z)$). It can be easily expanded to order 25 in a reasonable amount of time. This is the explicit truncated Bürmann series.

```
Bürmann[#^5 &, Sinh, {z, 0, 25}]
```

$$\mathrm{Sinh}[z]^5 - \frac{5\,\mathrm{Sinh}[z]^7}{6} + \frac{47\,\mathrm{Sinh}[z]^9}{72} - \frac{1571\,\mathrm{Sinh}[z]^{11}}{3024} + \frac{153\,617\,\mathrm{Sinh}[z]^{13}}{362\,880} - \frac{1\,206\,053\,\mathrm{Sinh}[z]^{15}}{3\,421\,440} + \frac{1\,447\,983\,367\,\mathrm{Sinh}[z]^{17}}{4\,843\,238\,400} - \frac{22\,449\,497\,227\,\mathrm{Sinh}[z]^{19}}{87\,178\,291\,200} + \frac{79\,923\,511\,502\,753\,\mathrm{Sinh}[z]^{21}}{355\,687\,428\,096\,000} - \frac{694\,675\,031\,171\,089\,\mathrm{Sinh}[z]^{23}}{3\,504\,179\,847\,168\,000} + \frac{2\,041\,637\,377\,789\,356\,133\,\mathrm{Sinh}[z]^{25}}{11\,563\,793\,495\,654\,400\,000}$$

The result is validated in terms of a Taylor series. This shows that the error is at least of order 26.

```
Series[Bürmann[#^5 &, Sinh, {z, 0, 25}], {z, 0, 25}]
```

$z^5 + O[z]^{26}$

```
Plot[Evaluate@{z^5, Bürmann[#^5 &, Sinh, {z, 0, 25}] },
 {z, -1, 1}, PlotStyle → {{Thick},
{Thick, Red, Dotted}}, FrameLabel -> {z, f},
PlotLabel →
  Row[{"Bürmann expansion at ", z₀, " = 0, ", m, " = ",
25}], Frame → True,
 PlotLegends →
  Placed[{TraditionalForm[z^5], "truncated series"},
   {.8, .2}]]
```




Here are the coefficients $\mathcal{B}_k$ as they are calculated with this procedure in their explicit analytic form, according to the definition given in (2).

```
B[f_, ϕ_, n_, z0_] := Module[{y}, ϕ′[z0]^n c[f, ϕ, n, y] /. y → z0]
```

```
TraditionalForm@
 TableForm[
  Table[Row[{ℬi, " = ", B[#^5 &, Sinh, i, z₀] // FactorTerms}],
   {i, 1, 3}]]
```

$\mathcal{B}_1 = 5\, z_0^4$

$\mathcal{B}_2 = -\frac{5}{2}\left(z_0^4 \tanh(z_0) - 4\, z_0^3\right)$

$\mathcal{B}_3 = -\frac{5}{6}\left(-12\, z_0^2 - 2\, z_0^4 \tanh^2(z_0) + 12\, z_0^3 \tanh(z_0) + z_0^4 \operatorname{sech}^2(z_0)\right)$

## □ The Recursive Formula for the Inversion Theorem

A useful application of the recursion (5)–(7) to the case of the expansion of the inverse function $\varphi(z)$ of $f(z)$ around $z = z_0$ follows immediately: we have $\varphi(f(z)) = z$ and it follows that the inverse function is the Bürmann series of $z$ in powers of $f(z) - f(z_0)$ (see [8]). By writing

$$\begin{aligned}
&\varphi(\zeta) - \varphi(\zeta_0) = \sum_{n=1}^{\infty} \mathcal{I}_n(z_0)\,(\zeta - \zeta_0)^n, \\
&\varphi(f(z)) - \varphi(f(z_0)) = \\
&\quad z - z_0 = \sum_{n=1}^{\infty} \mathcal{I}_n(z_0)\,(f(z) - f(z_0))^n = \sum_{n=1}^{\infty} \mathcal{C}_n(z; f, z_0)\,(f(z) - f(z_0))^n,
\end{aligned} \tag{8}$$

we obtain

$$\begin{aligned}
&\mathcal{I}_1(z_0) = \frac{1}{f'(z_0)}, \\
&\mathcal{I}_n(z_0) = \frac{1}{n}\,\frac{1}{f'(z_0)}\,\frac{d\mathcal{I}_{n-1}(z_0)}{dz_0}, \\
&\mathcal{I}_n(z_0) = \frac{1}{n!}\,\frac{1}{f'(z_0)}\,\frac{d}{dz_0}\left(\frac{1}{f'(z_0)}\,\frac{d}{dz_0}\left(\ldots\,\frac{1}{f'(z_0)}\,\frac{d}{dz_0}\left(\frac{1}{f'(z_0)}\right)_1\,\ldots\right)_{n-2}\right)_{n-1}.
\end{aligned} \tag{9}$$

The following program calculates the first three coefficients for the inverse function $f(z)$ in general, which corresponds to the expression shown in [8].

```
TraditionalForm@TableForm[Table[
   Row[{
     𝓘j, " = ", FactorTerms[
      c[InverseFunction[f], # &, j, x] /.
       InverseFunction[f][x] → y]
    }],
   {j, 1, 3}]]
```

$\mathcal{I}_1 = \frac{1}{f'(y)}$

$\mathcal{I}_2 = -\frac{f''(y)}{2\,f'(y)^3}$

$\mathcal{I}_3 = \frac{1}{6}\left(\frac{3\,f''(y)^2}{f'(y)^5} - \frac{f^{(3)}(y)}{f'(y)^4}\right)$

As an example, the inverse of the sine function is expanded, and the result is displayed for order 11.

```
Bürmann[InverseFunction[Sin], # &, {z, 0, 11}]
```

$$z + \frac{z^3}{6} + \frac{3\, z^5}{40} + \frac{5\, z^7}{112} + \frac{35\, z^9}{1152} + \frac{63\, z^{11}}{2816}$$

Compare it to the result of the *Mathematica* built-in function `InverseSeries` (see [9]).

```
InverseSeries[Series[Sin[x], {x, 0, 11}]]
```

$$x + \frac{x^3}{6} + \frac{3\, x^5}{40} + \frac{5\, x^7}{112} + \frac{35\, x^9}{1152} + \frac{63\, x^{11}}{2816} + O[x]^{12}$$

## □ Expanding a Function in Powers of Its First Derivative

If we choose the basis function $\phi(z)$ to be equal to the first derivative $f'(z)$ of $f(z)$, we find, by using formula (5), the recursive expression and the first three coefficients:

$$\begin{aligned}
&C_n(f;\, f',\, z_0) = \frac{1}{n}\, \frac{1}{f''(z_0)}\, \frac{d}{dz_0}\, C_{n-1}(f;\, f',\, z_0),\\
&\mathcal{B}_1(f;\, f',\, z_0) = f'(z_0),\\
&\mathcal{B}_2(f;\, f',\, z_0) = \frac{1}{2}\left(f''(z_0)^2 - f'(z_0)\, f'''(z_0) \,/\, f''(z_0)\right),\\
&\mathcal{B}_3(f;\, f',\, z_0) =\\
&\quad \frac{1}{6}\left(\frac{1}{f''(z_0)^2}\left(3\, f'(z_0)\, f'''(z_0)^2 - 2\, f''(z_0)^2\, f'''(z_0) - f'(z_0)\, f''(z_0)\, f^{\mathrm{IV}}(z_0)\right)\right).
\end{aligned} \tag{10}$$

The idea of expanding an analytic function using its derivative as a basis function is fruitful for cases where the function $f(z)$ is defined by an integral. It will be shown that solutions to linear and nonlinear problems of diffusion or heat transfer can be expressed as integrals. We get

$$f(z) = \int_{z_0}^{z} f'(\zeta)\, d\zeta \longrightarrow f(z) = \sum_{n=0}^{\infty} \mathcal{B}_n(f;\, f',\, z_0)\left(\frac{f'(z) - f'(z_0)}{f''(z_0)}\right)^n.$$

To illustrate the advantage of this technique, we choose the expansion of the transcendental function $f(z) = \ln(1+z)$. The function is defined by the integral

$$\begin{aligned}
&f(z) = \ln(1+z) = \int_0^z \frac{1}{1+\zeta}\, d\zeta \longrightarrow \phi(z) = f'(z) = \frac{1}{1+z},\\
&\phi'(z) = f''(z) = -\left(\frac{1}{1+z}\right)^2.
\end{aligned}$$

Using the results listed in (10) for the Bürmann coefficients, we arrive at once at the expansion around $z_0$:

$$f(z) = \ln(1+z_0) - (1+z_0)\left(\frac{1}{1+z} - \frac{1}{1+z_0}\right) +$$
$$\frac{1}{2}(1+z_0)^2\left(\frac{1}{1+z} - \frac{1}{1+z_0}\right)^2 - \frac{1}{3}(1+z_0)^3\left(\frac{1}{1+z} - \frac{1}{1+z_0}\right)^3 + \ldots$$
$$= \ln(1+z_0) + \sum_{n=1}^{\infty}\frac{1}{n}\left(\frac{z-z_0}{1+z}\right)^n.$$

Here is the series to order 11.

```
w[z_] := Bürmann[Log[1 + #] &, 1 / (1 + #) &, {z, 0, 11}]
```

```
w[z]
```

$$1 - \frac{1}{1+z} + \frac{1}{2}\left(-1+\frac{1}{1+z}\right)^2 - \frac{1}{3}\left(-1+\frac{1}{1+z}\right)^3 + \frac{1}{4}\left(-1+\frac{1}{1+z}\right)^4 -$$
$$\frac{1}{5}\left(-1+\frac{1}{1+z}\right)^5 + \frac{1}{6}\left(-1+\frac{1}{1+z}\right)^6 - \frac{1}{7}\left(-1+\frac{1}{1+z}\right)^7 + \frac{1}{8}\left(-1+\frac{1}{1+z}\right)^8 -$$
$$\frac{1}{9}\left(-1+\frac{1}{1+z}\right)^9 + \frac{1}{10}\left(-1+\frac{1}{1+z}\right)^{10} - \frac{1}{11}\left(-1+\frac{1}{1+z}\right)^{11}$$

It can be simplified.

```
w[z] /. 1 - 1/(1 + z) → Together[1 - 1/(1 + z)] /.
  -1 + 1/(1 + z) → Together[-1 + 1/(1 + z)]
```

$$\frac{z^{11}}{11\,(1+z)^{11}} + \frac{z^{10}}{10\,(1+z)^{10}} + \frac{z^9}{9\,(1+z)^9} + \frac{z^8}{8\,(1+z)^8} + \frac{z^7}{7\,(1+z)^7} +$$
$$\frac{z^6}{6\,(1+z)^6} + \frac{z^5}{5\,(1+z)^5} + \frac{z^4}{4\,(1+z)^4} + \frac{z^3}{3\,(1+z)^3} + \frac{z^2}{2\,(1+z)^2} + \frac{z}{1+z}$$

```
Plot[Evaluate@{Log[1 + z], w[z]}, {z, -0.7, 10},
PlotStyle → {{Thick}, {Thick, Red, Dotted}},
FrameLabel → {z, None}, PlotLabel →
Row[{"Bürmann expansion at ", Style["z", Italic]0,
    " = 0, ", Style["m", Italic], " = ", 25}],
Frame → True,
 PlotLegends →
  Placed[{TraditionalForm@Log[1 + z], "expansion"}, {.8, .2}]]
```




## The Generalized Form of Bürmann Series Based on a Combinatorial Approach

Although the representation given by the recursive formula in (5) is more efficient in terms of CPU time compared to formula (3), there is still the restriction of using basis functions with nonvanishing first derivative at the expansion position $z_0$, since $\phi'(z_0)$ appears in the denominator of the coefficients in (5)–(7). To overcome this limitation, we introduce an alternative representation of the Bürmann coefficients based on a combinatorial approach [10] that can be generalized.

Actually, the Bürmann expansions are related to Taylor series of reciprocal powers of analytic functions $\phi^*(z)$ represented by

$$\frac{1}{\phi^*(z)^m} = \frac{1}{\phi^*(z_0)^m} \sum_{k=0}^{\infty} \mathfrak{R}_k(\phi^*, m)\,(z - z_0)^k. \qquad (11)$$

The function $\phi^*(z)$ is uniquely determined by the basis function $\phi(z)$. We will show that explicit expressions for the expansion of $f(z)$ and the corresponding Bürmann coefficients $\mathcal{B}_n(f;\phi,z_0)$ as defined in (2) can be derived using the coefficients $\mathrm{R}_k(\phi^*,m)$. The Bürmann expansion in this representation reads as

$$\begin{aligned} f(z) &= f(z_0) + \sum_{n=1}^{\infty}\sum_{r=0}^{n-1} \frac{f^{(r+1)}(z_0)}{r!}\,\frac{1}{n}\,\mathrm{R}_{n-r-1}(\phi^*, n)\left(\frac{\phi(z)-\phi(z_0)}{\phi'(z_0)}\right)^n, \\ \phi^*(z) &= \sum_{n=0}^{\infty}\frac{\phi^{(n+1)}(z_0)}{(n+1)!}\,(z-z_0)^n. \end{aligned} \tag{12}$$

The formula for the Bürmann coefficient $\mathcal{B}_n(f;\phi,z_0)$ in terms of the coefficients $\mathrm{R}_k(\phi^*,m)$ is thus given by

$$\mathcal{B}_n(f;\phi,z_0) = \sum_{r=0}^{n-1}\frac{f^{(r+1)}(z_0)}{r!}\,\frac{1}{n}\,\mathrm{R}_{n-r-1}(\phi^*, n). \tag{13}$$

The explicit formulation of generalized Bürmann series using powers of functions with derivatives, vanishing up to the order $\nu$ at $z = z_0$, is given by

$$\begin{aligned} &f(z) = \\ &\quad f(z_0) + \sum_{n=1}^{\infty}\sum_{r=0}^{n-1}\frac{f^{(r+1)}(z_0)}{r!}\,\frac{1}{n}\,\mathrm{R}_{n-r-1}\left(\phi^*, \frac{n}{\nu+1}\right)\left(\sqrt[\nu+1]{\frac{(\nu+1)!\,(\phi(z)-\phi(z_0))}{\phi^{(\nu+1)}(z_0)}}\right)^n, \end{aligned} \tag{14}$$

and the expression for the Bürmann coefficient is

$$\mathcal{B}_n(f;\phi,z_0) = \sum_{r=0}^{n-1}\frac{f^{(r+1)}(z_0)}{r!}\,\frac{1}{n}\,\mathrm{R}_{n-r-1}\left(\phi^*, \frac{n}{\nu+1}\right). \tag{15}$$

Also, the special case of expanding the inverse function $\varphi(z)$ of $f(z)$ can be derived in this general way, and one gets

$$\begin{aligned} &\varphi(z) = \\ &\quad z_0 + \sum_{n=1}^{\infty}\frac{1}{n}\,\mathrm{R}_{n-1}\left(f^*, \frac{n}{\nu+1}\right)\left(\left(((\nu+1)!\,(z-f(z_0)))\Big/ f^{(\nu+1)}(z_0)\wedge\left(\frac{1}{\nu+1}\right)\right)\right)^n, \\ &\qquad f^*(z) = \sum_{k=0}^{\infty}\frac{f^{(\nu+1+k)}(z_0)}{(\nu+1+k)!}\,(z-z_0)^k. \end{aligned} \tag{16}$$

The standard case (i.e. $\phi'(z_0) \neq 0$ and $f'(z_0) \neq 0$) is obtained by setting $\nu = 0$ in (14)–(16). By using the combinatorial approach as shown in the next subsection, one can also evaluate the coefficients in expansions resulting from the theorem of Teixeira [5], which is a generalization of Bürmann's theorem to singular functions.

The approach is explained in more detail and demonstrated with examples coded in *Mathematica* in the following.

## □ Explicit Expressions for the Coefficients of the Taylor Expansion of $f(z)^{-m}$ and $f(z)^{-\mu/\nu}$ Using Partitions

A partition of the positive integer $k$ is a sequence of positive integers $(j_1, j_2, \ldots, j_n)$ with $j_1 \geq j_2 \geq \ldots \geq j_n$ such that $j_1 + j_2 + \ldots + j_n = k$, for example, $(4, 3, 2, 1, 1, 1)$, a partition of 10 that is usually written as $4+3+2+1+1+1$. The number of times a part $s \in \{1, 2, \ldots, k\}$ occurs in a partition $p$ is its frequency $p_s$. In the example, the parts $1, 2, 3, 4$ occur with frequencies $3, 1, 1, 1$; the example partition can be written as $1^3\, 2^1\, 3^1\, 4^1$.

```
Last /@ Sort[Tally[{4, 3, 2, 1, 1, 1}]]
```

```
{3, 1, 1, 1}
```

For a partition $p$ of $k$, define the vector of frequencies, $\text{freq}(p) = (p_1, p_2, \ldots, p_k)$. Then

$$\sum_{s=1}^{k} s\, p_s = k; \tag{17}$$

for convenience, define

$$p_0 = k - \sum_{s=1}^{k} p_s. \tag{18}$$

In the example, $\text{freq}(4, 3, 2, 1, 1, 1) = (3, 1, 1, 1, 0, 0, 0, 0, 0, 0)$ and $p_0 = 10 - (3 + 1 + 1 + 1) = 4$.

In *Mathematica*, `IntegerPartitions[k]` gives all possible partitions of $k$.

```
IntegerPartitions[4]
```

```
{{4}, {3, 1}, {2, 2}, {2, 1, 1}, {1, 1, 1, 1}}
```

The frequencies of the parts, $p_s$, can be found with `FrobeniusSolve[Range[k], k]`.

```
FrobeniusSolve[Range[4], 4]
```

```
{{0, 0, 0, 1}, {0, 2, 0, 0},
 {1, 0, 1, 0}, {2, 1, 0, 0}, {4, 0, 0, 0}}
```

However, `FrobeniusSolve` is slow for integers larger than about 30.

```
Timing[FrobeniusSolve[Range[30], 30];]
```

```
{2.437500, Null}
```

The function `PartitionsM`, based on `IntegerPartitions`, is significantly faster.

```
MultiplicitiesFromParts[p_] :=
  Count[p, #] & /@ Range[Total@p];

PartitionsM[p_Integer] :=
  MultiplicitiesFromParts /@ IntegerPartitions[p];

PartitionsM[4]

{{0, 0, 0, 1}, {1, 0, 1, 0},
 {0, 2, 0, 0}, {2, 1, 0, 0}, {4, 0, 0, 0}}

Timing[PartitionsM[30];]

{0.314763, Null}
```

The function `PartitionsJ`, based on an undocumented but highly efficient function [11], is even faster.

```
PartitionsJ[p_] :=
 Reduce`NaturalLinearSolve[{Range[p]}, {p}, True][[1]]

PartitionsJ[4]

{{4, 0, 0, 0}, {2, 1, 0, 0},
 {0, 2, 0, 0}, {0, 0, 0, 1}, {1, 0, 1, 0}}

Timing[PartitionsJ[30];]

{0.023015, Null}
```

According to

$$\frac{1}{f(z)^m} = \frac{1}{f(z_0)^m} \sum_{k=0}^{\infty} \mathfrak{R}_k(f, m)\,(z - z_0)^k, \tag{19}$$

the coefficients $\mathfrak{R}_k(f, m)$ of reciprocal powers of analytic functions $f(z)$ can be derived explicitly on the basis of combinatorics and analysis, recapitulated in appendix A. When $m$ is an integer, the coefficients are

$$\mathcal{R}_k(f, m) =$$
$$\sum_{\substack{\text{all partitions } p \text{ of } k, \\ \text{freq}(p)=(p_1,\ldots,p_k)}} (-1)^{(k-p_0)} (m+k-p_0-1)!\,/\,(m-1)! \prod_{s=1}^{k} \left\{ \frac{1}{p_s!} \left( \frac{f^{(s)}(z_0)}{s!\, f(z_0)} \right)^{p_s} \right\}. \tag{20}$$

The case when $m$ is rational, $m = \mu / \nu$, is relevant for Bürmann expansion with basis functions with vanishing derivatives $\phi^{(1,2,..,\nu)}(z_0)$. In that case,

$$\mathcal{R}_k\left(f, \frac{\mu}{\nu}\right) = \sum_{\substack{\text{all partitions } p \text{ of } k, \\ \text{freq}(p)=(p_1,\ldots,p_k)}} (-1)^{(k-p_0)} \prod_{r=0}^{k-p_0-1} \left(\frac{\mu}{\nu} + r\right) \prod_{s=1}^{k} \frac{1}{p_s!} \left( \frac{f^{(s)}(z_0)}{s!\, f(z_0)} \right)^{p_s}. \tag{21}$$

Now we show how to calculate (21) symbolically. For example, choosing the function $f(z) = 1 + 7\,z + 8\,z^2 + 11\,z^3$, let us calculate $\frac{1}{f(z)^{3/5}}$.

We use the fact that the *Mathematica* functions `Times`, `Plus`, and `Total` work with empty lists.

```
{Times @@ {}, Plus @@ {}, Total[{}]}
```

```
{1, 0, 0}
```

To avoid the undefined expression 0^0, the differentiation is performed analytically first on the symbolic function $g$. Then raising to the power of $p_s$ (which can be zero) is performed, and finally the symbolic function $g$ is substituted out by the function $f$.

```
Rr[f_, z0_, μ_, ν_, k_] := Module[
  {g},
  Total[
    Map[(-1)^Total[#] (Product[(μ / ν + r), {r, 0, Total[#] - 1}]) 1/(Times @@ (#!))
       Times @@ MapIndexed[
          (Derivative[First[#2]][g][z0] / (First[#2]! * g[z0]))^#1 &, #] &,
PartitionsJ[k]]] /. g → f
  ]
```

Here are the $\mathfrak{R}_k(f, 3/5)$ for the symbolic function $F(z)^{3/5}$ expanded at $\zeta_0$.

```
TraditionalForm@
 TableForm[Table[Row[{ℜk, " = ", ℜr[F, ζ0, 3, 5, k]}],
   {k, 0, 3}]]
```

$$\mathfrak{R}_0 = 1$$
$$\mathfrak{R}_1 = -\frac{3 F'(\zeta_0)}{5 F(\zeta_0)}$$
$$\mathfrak{R}_2 = \frac{12 F'(\zeta_0)^2}{25 F(\zeta_0)^2} - \frac{3 F''(\zeta_0)}{10 F(\zeta_0)}$$
$$\mathfrak{R}_3 = -\frac{F^{(3)}(\zeta_0)}{10 F(\zeta_0)} - \frac{52 F'(\zeta_0)^3}{125 F(\zeta_0)^3} + \frac{12 F'(\zeta_0) F''(\zeta_0)}{25 F(\zeta_0)^2}$$

Now we expand $f(z)^{3/5}$ at $z_0 = 0.2$. For convenience, define `auxf`.

```
auxf[f_, z_, z0_, μ_, ν_, n_] :=
 1 / f[z0] ^ (μ / ν) Sum[ℜr[f, z0, μ, ν, k] (z - z0) ^k, {k, 0, n}]
```

```
Module[{f, μ, ν, z0},
 f[z_] := 1 + 7 z + 8 z^2 + 11 z^3;
 μ = 3;
 ν = 5;
 z0 = 0.2;
 Plot[
  Evaluate[{1 / f[z] ^ (μ / ν), auxf[f, z, z0, μ, ν, 2],
    auxf[f, z, z0, μ, ν, 6], auxf[f, z, z0, μ, ν, 10]}],
  {z, z0 - 0.25, z0 + 0.25},
  Frame → True,
  FrameLabel → {z, Style["f", Italic]},
  PlotLabel → "Expansion of 1/f^3/5 at z0 = 0.2",
  PlotStyle → {Black, Dashed, Dashed, Dashed},
  PlotLegends → Placed[{
     Row[{"1/", Style["f", Italic]^3/5}],
     Row[{"series (", Style["n", Italic], " = 2)"}],
     Row[{"series (", Style["n", Italic], " = 6)"}],
     Row[{"series (", Style["n", Italic], " = 10)"}]
    }, {.8, .7}],
  PlotRange → All
 ]
]
```




While the expansion is valid for complex-valued functions, the plot shows only the real part of $f$.

## □ The Explicit Expression for the Coefficients $\mathcal{B}_n(f;\phi, z_0)$ in Terms of the Coefficients $\mathrm{R}_k(\phi^*, m)$

Explicit expressions for the Bürmann coefficients $\mathcal{B}_n(f;\phi, z_0)$ as they are defined in (2) can be defined with respect to the coefficients $\mathrm{R}_k(\phi^*, m)$.

Again using `FrobeniusSolve[Range[k], k]` (see appendix A), we can formulate the general expressions for Bürmann series using functions $\phi(z)$ with vanishing derivatives $\phi^{(1,2,\ldots,\nu)}(z_0)$. For instance, series of powers of functions of this type can give convergent expansions for functions that are defined by integrals, like the error function

$$\operatorname{erf}(z) = \frac{2}{\sqrt{\pi}} \int_0^z e^{-\zeta^2}\, d\zeta, \tag{22}$$

which plays a key role in the theory of linear and nonlinear heat transfer [1]. Defining the integrand as the basis function of a Bürmann series, as explained in (10), we will find a rapidly converging series representation of $\operatorname{erf}(z)$.

## □ Expressions for the Bürmann Coefficients $\mathcal{B}_n(f;\phi, z_0)$

All the results of the previous section can be applied to get a formula for the Bürmann coefficients that is efficiently implemented in a simple function. The starting point of the derivation of this expression is a formulation of the Bürmann expansion in terms of a complex contour integral, as it is given in [5]. This approach can be found in various presentations [6, 7]. The evaluation of the integral representation of the Bürmann expansion results in

$$f(z) = f(z_0) + \sum_{n=1}^{\infty}\sum_{r=0}^{n-1} \frac{f^{(r+1)}(z_0)}{r!}\,\frac{1}{n}\,\mathfrak{R}_{n-r-1}(\phi^*, n)\left(\frac{\phi(z)-\phi(z_0)}{\phi'(z_0)}\right)^n,$$
$$\phi^*(z) = \sum_{n=0}^{\infty} \frac{\phi^{(n+1)}(z_0)}{(n+1)!}\,(z-z_0)^n. \tag{23}$$

The formula for the Bürmann coefficient $\mathcal{B}_n(f;\phi, z_0)$ in terms of the coefficients $\mathfrak{R}_k(\phi^*, m)$ is thus given by

$$\mathcal{B}_n(f;\phi, z_0) = \sum_{r=0}^{n-1} \frac{f^{(r+1)}(z_0)}{r!}\,\frac{1}{n}\,\mathfrak{R}_{n-r-1}(\phi^*, n). \tag{24}$$

The function `fbür` shows how to apply (23) and (24) to the expansion of $f(z) = z^5$ in powers of $\phi(z) = \sinh(z)$, the same example as presented in the first section. The series is expanded up to order 15, so that the error is at least of order 16.

```
Rbür[fun_, z0_, m_, n_] :=
 Total[Map[(-1)^(Plus@@#) * ((m - 1 + (Plus @@ #)) !)/((m - 1) ! Times @@ (# !)) *
Times @@ MapIndexed[(Derivative[First[#2] + 1][g][z0]/((First[#2] + 1) ! * g'[z0]))^#1 &,
          #] &,
PartitionsJ[n]]] /. g → fun

ℬ[fun_, bfun_, n_, z0_] :=
 Sum[(Derivative[r + 1][fun][ξ]/r ! Rbür[bfun, z0, n, n - r - 1]/n), {r, 0, n-1}] /.
  ξ → z0

fbür[fun_, bfun_, z_, z0_, n_] :=
 fun[z0] + Sum[ℬ[fun, bfun, j, z0] ((bfun[z] - bfun[z0])/(bfun'[z0]))^j, {j, 1, n}]
```

```
Module[{f, ϕ, mexp},
 f[z_] := z^5; (* function to expand *)
 ϕ[z_] := Sinh[z]; (* basis function for expansion *)
 mexp = 15; (* order of expansion *)
 z0 = 0; (* center of series expansion *)
 Column[{
   Text@TraditionalForm@
     Row[{"Function to expand: ", f[z]}],
   Text@TraditionalForm@
     Row[{"Basis function ϕ(", Style["z", Italic],
       ") for expansion: ", ϕ[z]}],
   Text@TraditionalForm@
     Row[{"The order of expansion is chosen by: ",
       Style["m", Italic], " = ", mexp}],
   Text@TraditionalForm@
     Row[{"Bürmann expansion of ", f[z], " at ",
       Style["z", Italic]_0, " = ", z0, ":"}],
   Text@TraditionalForm@fbür[f, ϕ, z, z0, mexp],
   (* explicit expression of the Bürmann series *)
   Text@TraditionalForm@
     Row[
      {"The error of the expansion is at least in
         the order of ",
       Series[fbür[f, ϕ, z, z0, mexp], {z, 0, 15}] - f[z]}]
  }]
(* showing that the error is at least in the order of mexp+
  1 *)
]
```

Function to expand: $z^5$

Basis function $\phi(z)$ for expansion: $\sinh(z)$

The order of expansion is chosen by: $m = 15$

Bürmann expansion of $z^5$ at $z_0 = 0$:

$$-\frac{1\,206\,053 \sinh^{15}(z)}{3\,421\,440} + \frac{153\,617 \sinh^{13}(z)}{362\,880} - \frac{1571 \sinh^{11}(z)}{3024} + \frac{47 \sinh^9(z)}{72} - \frac{5 \sinh^7(z)}{6} + \sinh^5(z)$$

The error of the expansion is at least in the order of $O\left(z^{16}\right)$

### Application to the Inversion Theorem

For the special case of the inverse function $\varphi(z)$ of $f(z)$ given by (12), in using the transformations indicated in (8),

$$\mathcal{I}_n(f, z_0) = C_n(z; f, z_0) = \frac{\mathfrak{R}_{n-1}(f^*, n)}{n f'(z_0)^n}, f^*(z) = \sum_{n=0}^{\infty} \frac{f^{(n+1)}(z_0)}{(n+1)!} (z - z_0)^n. \tag{25}$$

So the expansion of the inverse function can be expressed as

$$\varphi(z) = z_0 + \sum_{n=0}^{\infty} \frac{\mathfrak{R}_{n-1}(f^*, n)}{n} \left( \frac{z - f(z_0)}{f'(z_0)} \right)^n. \tag{26}$$

Equation (26) is the compact formulation of a result given by Morse and Feshbach [12]. As a result of our approach, we have developed formulas for Bürmann coefficients and for the expansion coefficients of inverse functions that reveal the close relationship of these coefficients to the coefficients for reciprocal powers of an analytic function defined in (20). In the following section, we present a generalization of Bürmann's theorem, using the solutions of equation (17).

## Bürmann Series with Functions for Which the First $\nu$ Derivatives Vanish

Inspecting formulas (23) and (26), we notice that they cannot be evaluated for cases where $\phi'(z_0)$ or $f'(z_0)$ vanishes. This shortcoming must be overcome, for in some cases of interest we will be forced to find Bürmann series using basis functions whose first $\nu$ derivatives at $z_0$ vanish:

$$\phi(z) = \phi(z_0) + \sum_{n=\nu+1}^{\infty} \frac{\phi^{(n)}(z_0)}{n!} (z - z_0)^n, \phi^{(\nu+1)}(z_0) \neq 0, \nu \geq 1. \tag{27}$$

### Change to a Multivalued Function

To this end, define the multivalued function

$$\Theta(\phi, z_0, z) = \sqrt[\nu+1]{\phi(z) - \phi(z_0)}\,, \tag{28}$$

which is cast into the form

$$\Theta(\phi, z_0, z) = (z - z_0) \sqrt[\nu+1]{\phi^*(z)}\,, \phi^*(z) = \sum_{k=0}^{\infty} \frac{\phi^{(\nu+1+k)}(z_0)}{(\nu+1+k)!} (z - z_0)^k. \tag{29}$$

The function $\phi^*(z)^{1/(\nu+1)}$ in (29) can be expanded into a Taylor series with $\phi^{*\prime}(z_0) \neq 0$, and hence (28) fulfills the condition violated by $\phi(z)$. Thus, instead of expanding $f(z)$ in powers of $\Phi(\phi, z_0, z)$, we expand in powers of $\Theta(\phi, z_0, z)$. A reformulation of the contour

integral [5] results in the generalized form of Bürmann's theorem given in (14). Actually, the introduction of the root function (28) in (14) leads to several solution branches. For real-valued functions $f(z)$, the use of the `Sign` and `Abs` functions in the following code extracts the correct branch of the root function for numerical purposes. For a formal proof of the equivalence of the Bürmann series and the Taylor series for $f(z)$, the `Sign` and `Abs` functions can be omitted.

## ■ *The Generalized Form of Bürmann Series*

As an example, we calculate an expansion of arcsin[$z$] up to order 15 according to (14) in powers of $\left(\sqrt{1-z^2}\right)^{-1}$, a basis function with a vanishing first derivative at $z = 0$ (i.e. $\nu = 2$):

```
Rνbür[bfun_, z0_, μ_, ν_, n_] :=
 Total[Map[(-1)^(Plus@@#) (∏_{r=0}^{Plus@@#-1} (μ/ν + r))
 1/(Times @@ (#!))
      Times @@ MapIndexed[
        ((ν! * Derivative[First[#2] + ν][g][z0]) / ((First[#2] + ν)! * Derivative[ν][g][z0]))^#1 &, #] &,
 PartitionsJ[n]]] /. g → bfun

𝓑ν[fun_, bfun_, n_, z0_, ν_] :=
 Σ_{r=0}^{n-1} (Derivative[r + 1][fun][ξ] / r!
      Rνbür[bfun, z0, n, ν, n - r - 1] / n) /. ξ → z0
```

Use this definition for the formal proof of equivalence.

```
fνbürtest[fun_, bfun_, z_, z0_, n_, ν_] := fun[z0] +
 Σ_{j=1}^{n} 𝓑ν[fun, bfun, j, z0, ν] ((ν! * (bfun[z] - bfun[z0])) / Derivative[ν][bfun][z0])^(j/ν)
```

Use this definition for numerical purposes, such as for plotting.

```
fνbür[fun_, bfun_, z_, z0_, n_, ν_] := fun[z0] +
 Sum[ℬν[fun, bfun, j, z0, ν] Sign[z - z0]^j
   (Abs[(ν! (bfun[z] - bfun[z0]))/(Derivative[ν][bfun][z0])])^(j/ν), {j, 1, n}]

Module[{f, ϕ, mexp, z0, ν0},
 f[z_] := ArcSin[z]; (* function to expand *)
 ϕ[z_] := 1/Sqrt[1 - z^2]; (* basis function for expansion *)
 mexp = 15; (* order of expansion *)
 z0 = 0; (* center of series expansion *)
 ν0 = 1;
 While[Derivative[ν0][ϕ][z0] == 0, ν0++];
 (* automatic detection of
 order of first nonvanishing derivative of ϕ at z=z0 *)

 Column[{
   Text@TraditionalForm@
     Row[{"Function to expand: ", f[z]}],
   Text@TraditionalForm@
     Row[{"Basis function ϕ(", Style["z", Italic],
       ") for expansion: ", ϕ[z]}],
   Text@TraditionalForm@
     Row[
      {"The order ν+1 of first nonvanishing derivative
          of ϕ(", Style["z", Italic], ") at z₀ is ",
       ν0}],
   Text@TraditionalForm@
     Row[{"The order of expansion is chosen by: ",
       Style["m", Italic], " = ", mexp}],
   Text@TraditionalForm@
     Row[{"Bürmann expansion of ", f[z], " at ",
       Style["z", Italic]₀, " = ", z0, ":"}],
   Text@TraditionalForm@fνbür[f, ϕ, z, z0, mexp, ν0],
   (* explicit expression of the Bürmann series *)
   Text@TraditionalForm@
     Row[
      {"The error of the expansion is at least in
          the order of ",
       Series[fνbürtest[f, ϕ, z, z0, mexp, ν0], {z, 0, 15}] -
        f[z]}],
   (* showing that the error is at least in the
     order of mexp+1 *)
```

```
    "",
    Plot[
     Prepend[Table[fνbür[f, ϕ, z, z0, i, ν0],
        {i, {2, 5, 10, 15}}],
       f[z]] // Evaluate, {z, -1, 1},
     PlotRange → {-π / 2, π / 2},
     Frame → True, FrameLabel → {z, Style["f", Italic]},
     PlotLegends → Placed[{
        Style["f", Italic],
        Row[{Style["f", Italic]"Bür", " (",
          Style["n", Italic], " = 2)"}],
        Row[{Style["f", Italic]"Bür", " (", Style["n", Italic],
          " = 5)"}],
        Row[{Style["f", Italic]"Bür", " (", Style["n", Italic],
          " = 10)"}],
        Row[{Style["f", Italic]"Bür", " (", Style["n", Italic],
          " = 15)"}]
       }, {.7, .25}],
     PlotStyle → {Black, Dashed, Dashed, Dashed, Dashed},
     PlotLabel →
      Row[{"Expansion of ", ArcSin[z], "  at ", z0,
        " = ", z0}],
     ImageSize → 400
    ]
}]
```

Function to expand: $\sin^{-1}(z)$

Basis function $\phi(z)$ for expansion: $\frac{1}{\sqrt{1-z^2}}$

The order $\nu+1$ of first nonvanishing derivative of $\phi(z)$ at $z_0$ is 2

The order of expansion is chosen by: $m = 15$

Bürmann expansion of $\sin^{-1}(z)$ at $z_0 = 0$:

$$-\frac{74069 \left|\frac{1}{\sqrt{1-z^2}}-1\right|^{15/2} \operatorname{sgn}(z)^{15}}{393216\sqrt{2}} + \frac{92479 \left|\frac{1}{\sqrt{1-z^2}}-1\right|^{13/2} \operatorname{sgn}(z)^{13}}{425984\sqrt{2}} - \frac{11531 \left|\frac{1}{\sqrt{1-z^2}}-1\right|^{11/2} \operatorname{sgn}(z)^{11}}{45056\sqrt{2}} + \frac{2867 \left|\frac{1}{\sqrt{1-z^2}}-1\right|^{9/2} \operatorname{sgn}(z)^{9}}{9216\sqrt{2}} - \frac{177 \left|\frac{1}{\sqrt{1-z^2}}-1\right|^{7/2} \operatorname{sgn}(z)^{7}}{448\sqrt{2}} + \frac{43 \left|\frac{1}{\sqrt{1-z^2}}-1\right|^{5/2} \operatorname{sgn}(z)^{5}}{80\sqrt{2}} - \frac{5 \left|\frac{1}{\sqrt{1-z^2}}-1\right|^{3/2} \operatorname{sgn}(z)^{3}}{6\sqrt{2}} + \sqrt{2}\sqrt{\left|\frac{1}{\sqrt{1-z^2}}-1\right|}\operatorname{sgn}(z)$$

The error of the expansion is at least in the order of $O\left(z^{16}\right)$




A faster convergence can be achieved by using a basis function of the form $\phi(z) = z^3 \Big/ \sqrt{1 - z^2}$, for which the first and second derivatives vanish at the point $z_0 = 0$ (i.e. $\nu + 1 = 3$).

## ■ *Generalization of the Inversion Theorem*

Using formula (14), it is easy to deduce the expansion of the inverse function of an analytic function of the form

$$f(z) = f(z_0) + \sum_{n=\nu>1}^{\infty} \frac{f^{(n)}(z_0)}{n!} (z - z_0)^n. \tag{30}$$

The inverse function $\varphi(z)$ of $f(z)$ comes from (14) by setting $f(z) \to z$ and $\phi(z_0) \to f(z_0)$:

$$\begin{aligned} \varphi(z) &= z_0 + \sum_{n=1}^{\infty} \frac{1}{n} \mathfrak{R}_{n-1}\left(f^*, \frac{n}{\nu+1}\right) \left( \sqrt[\nu+1]{\frac{(\nu+1)!\,(z - f(z_0))}{f^{(\nu+1)}(z_0)}} \right)^n, \\ f^*(z) &= \sum_{k=0}^{\infty} \frac{f^{(\nu+1+k)}(z_0)}{(\nu+1+k)!} (z - z_0)^k. \end{aligned} \tag{31}$$

## Expanding the Error Function erf($z$)

The error function

$$\operatorname{erf}(z) = \frac{2}{\sqrt{\pi}} \int_0^z e^{-\zeta^2}\, d\zeta$$

is the first example demonstrating the efficiency of Bürmann series using the first derivative as a basis function. We define the function $f(z)$ and the basis function $\phi(z)$ by

$$f(z) = \frac{\sqrt{\pi}}{2} \operatorname{erf}(z) = \int_0^z e^{-\zeta^2}\, d\zeta \longrightarrow \phi(z) = f'(z) = e^{-z^2},$$
$$\phi'(z) = f''(z) = -2\, z\, e^{-z^2}.$$

The error function will be expanded around the origin $z_0 = 0$, where we find that $\phi'(z_0) = 0$. This expansion thus calls for the application of the generalized form of the Bürmann expansion given in (14). Hence we have to set, according to (28),

$$\Theta(\phi, 0, z) = \sqrt{1 - e^{-z^2}}\,.$$

To evaluate (14), we use the following relations for the derivatives of the integrand:

$$\phi^{(2\,n)}(0) = 2\,(-1)^n \frac{(2\,n-1)!}{(n-1)!},$$
$$\phi^{(2\,n+1)}(0) = 0.$$

The result of this calculation performed up to order nine in $\Theta\,(\phi, 0, z)$ is

$$\operatorname{erf}(z) = \frac{2}{\sqrt{\pi}} \left(\Theta(\phi, 0, z) - \frac{1}{12}\Theta(\phi, 0, z)^3 - \frac{7}{480}\Theta(\phi, 0, z)^5 - \frac{5}{896}\Theta(\phi, 0, z)^7 - \frac{787}{276\,480}\Theta(\phi, 0, z)^9 - \ldots\right). \tag{32}$$

A function calculating the expansion (32) is given below. To show that this approach is superior to a common Taylor expansion in a plot, we calculate the power series in $z$ up to order 10.

```
Module[{f, ϕ, mexp, z0, ν0, erfB, erfT},
 f[z_] := Erf[z]; (* function to expand *)
 ϕ[z_] = Sqrt[π]/2 D[f[z], z];
 (* basis function for expansion *)
 mexp = 10; (* order of expansion *)
 z0 = 0; (* center of series expansion *)
 ν0 = 1; While[Derivative[ν0][ϕ][z0] == 0, ν0++];
```

```
(* automatic detection of order of 1st nonvanishing
  derivative ν0 of ϕ at z=z0 *)
erfB[z_] = fνbür[f, ϕ, z, z0, mexp, ν0];
(* explicit expression of the Bürmann series *)
erfT[z_] = Series[Erf[z], {z, z0, mexp}] // Normal;
(* Taylor series *)
Column[
 {Text@TraditionalForm@
    Row[{"Function to expand: ", f[z]}],
  Text@TraditionalForm@
    Row[{"Basis function ϕ(", Style["z", Italic],
      ") for expansion: ", ϕ[z]}],
  Text@TraditionalForm@
    Row[
     {"The order ν+1 of first nonvanishing derivative
        of ϕ(", Style["z", Italic], ") at ",
      Style["z", Italic]0, " is ", ν0}],
  Text@TraditionalForm@
    Row[{"The order of expansion is chosen by: ",
      Style["m", Italic], " = ", mexp}],
  Text@TraditionalForm@
    Row[{"Bürmann expansion of ", f[z], " at ",
      Style["z", Italic]0, " is:"}],
  Text@TraditionalForm@erfB[z],
  "",
  Plot[{f[z], erfB[z], erfT[z]}, {z, -1, 5},
   PlotRange → {-1, π / 2},
   Frame → True, FrameLabel → {z, Style["f", Italic]},
   PlotLegends → Placed[{
      Row[{"erf(", Style["z", Italic], ")"}],
      Row[{Style["f", Italic]"Bür", " (", Style["n", Italic],
        " = 10)"}],
      Row[{Style["f", Italic]"Taylor", " (",
        Style["n", Italic], " = 10)"}]
     }, {.8, .2}],
   PlotStyle → {Black, {Red, Dashed}, {Dashed, Blue}},
   PlotLabel →
    Row[{"Expansion of ", Erf[z], "  at ", z0, " = ", z0}],
   ImageSize → 400]
 }]
]
```

Function to expand: erf($z$)

Basis function $\phi(z)$ for expansion: $e^{-z^2}$

The order $\nu$+1 of first nonvanishing derivative of $\phi(z)$ at $z_0$ is 2

The order of expansion is chosen by: $m$ = 10

Bürmann expansion of erf($z$) at $z_0$ is:

$$-\frac{787\left|1-e^{-z^2}\right|^{9/2}\operatorname{sgn}(z)^9}{138\,240\sqrt{\pi}}-\frac{5\left|1-e^{-z^2}\right|^{7/2}\operatorname{sgn}(z)^7}{448\sqrt{\pi}}-\frac{7\left|1-e^{-z^2}\right|^{5/2}\operatorname{sgn}(z)^5}{240\sqrt{\pi}}-\frac{\left|1-e^{-z^2}\right|^{3/2}\operatorname{sgn}(z)^3}{6\sqrt{\pi}}+\frac{2\sqrt{\left|1-e^{-z^2}\right|}\operatorname{sgn}(z)}{\sqrt{\pi}}$$




The plot shows that the series in (32) has only a small constant offset error for larger values of $z$, whereas the Taylor expansion dramatically deviates for smaller values of $z$, although it converges uniformly for all values of $z$. The series in (32) converges uniformly for all $z$ and gives the exact value for the error function. The rearrangement of terms, leading to $\mathrm{erf}_B(z)$, is thus justified. Even for the lowest order, we will find a result that shows no unbounded error, unlike the Taylor series.

```
Module[{f, ϕ, mexp, z0, ν0, erfB, erfT},
 f[z_] := Erf[z];
 ϕ[z_] = (√π / 2) D[f[z], z];
 (* basis function for expansion *)
 mexp = 10; (* order of expansion *)
 z0 = 0; (* center of series expansion *)
 ν0 = 1; While[Derivative[ν0][ϕ][z0] == 0, ν0++];
 erfB[z_] = fνbür[f, ϕ, z, z0, mexp, ν0];
 (* explicit expression of the Bürmann series *)
 erfT[z_] = Series[Erf[z], {z, z0, mexp}] // Normal;
```

```
LogPlot[{f[z] - erfB[z], f[z] - erfT[z]} // Abs // Evaluate,
  {z, 0.2, 5}, PlotRange → {1*^-8, 1*^4},
 Frame → True, FrameLabel → {"z", "error"},
  PlotLegends → Placed[{
     Row[{Style["f", Italic]"Bür", " (", Style["n", Italic],
       " = 10)"}],
     Row[{Style["f", Italic]"Taylor", " (", Style["n", Italic],
       " = 10)"}]
    }, {.8, .2}],
 PlotStyle → {{Red, Dashed}, {Dashed, Blue}},
  PlotLabel →
   Row[{"Error of expansion of ", Erf[z], " at ", z0,
     " = ", 0}],
  ImageSize → 400]
]
```




Due to the uniform convergence of (32), we can write:

$$\operatorname{erf}(z) = \frac{2\,\operatorname{sign}(z)}{\sqrt{\pi}} \sqrt{1 - e^{-z^2}} \left(c_0 + c_1\, e^{-z^2} + c_2\, e^{-2 z^2} + \ldots\right). \tag{33}$$

Using $\lim_{z\to\infty} \operatorname{erf}(z) = 1$ and $\lim_{z\to\infty} e^{-k z^2} = 0$ in (33), we find $c_0 = \sqrt{\pi} \big/ 2$. So by reordering the sums, we automatically get rid of the offset error at infinity. In fact, we can furthermore achieve a practical application of (33) by keeping only a few coefficients $c_i$. For example, using only $c_1$ and requesting the correct slope at $z = 0$, one gets an approximation of the error function with a relative error smaller than 1.2%. Taking additional terms of (33) with meaningful conditions further improves the approximating series in

(33), as also shown in the following plot (choosing $c_1 = \frac{31}{200}$ and $c_2 = \frac{11}{40} c_1$).

```
Module[
 {erfc1, sol1, erfc1c2, c1, c2},
 erfc1[c1_, z_] = Sign[z] 2/Sqrt[π] Sqrt[1 - Exp[-z^2]] *
   (Sqrt[π]/2 + c1 Exp[-z^2]);
 (* series according to (33) with a₀ and a₁ *)
 sol1 =
  Solve[(D[Erf[z], z] /. z → 0) ==
     Limit[D[erfc1[c1, z], z], z → 0], c1][[1]];
 (* calculation of a₁ by requesting correct slope at z=
  0 *)
 erfc1c2[c1_, c2_, z_] =
  Sign[z] 2/Sqrt[π] Sqrt[1 - Exp[-z^2]]
   (Sqrt[π]/2 + c1 Exp[-z^2] + c2 Exp[-2 z^2]);
 (* series according to (33) with a₀, a₁, and a₂ *)
 Plot[{(Erf[z] - erfc1[c1 /. sol1, z]) / Erf[z],
   (Erf[z] - erfc1c2[31 / 200, -11 / 40 * 31 / 200, z]) / Erf[z]},
  {z, -3, 3}, Frame → True,
  FrameLabel → {z, "relative error"},
  PlotLegends → Placed[
    {
     Row[{Style["c", Italic]₁, " = ", c1 /. sol1, " ≈ ",
        N[c1 /. sol1]}],
     Row[{Style["c", Italic]₁, " = ", 31/200, ", ",
        Style["c", Italic]₂, " = ", -11/40,
        Style["c", Italic]₁}]
    }, Below]
 ]
]
```




## ■ Application of Bürmann Series to Problems of Nonlinear Heat Transfer

This section applies the concept of Bürmann series to get solutions of nonlinear differential equations. After a short introduction, an example from the field of the diffusion type is presented.

### □ Application to Ordinary Differential Equations

In studying nonlinear ordinary differential equations, we cannot, in general, expect to find an exact solution expressible in terms of commonly used algebraic or transcendental functions. This difficulty is illustrated by the equations studied by Fujita, Lee, and Crank, which we will encounter later [13–17]. For the case of a general nonlinear second-order equation

$$f''(z) = h(z, f', f),\ f(z_0) = f_0,\ f'(z_0) = g_0 \neq 0, \tag{34}$$

where $h(z, f', f)$ denotes an analytic function of its arguments, one approach is to cast a solution into the form of a series of powers of the independent variable $z$. Depending on the complexity of the expression $h(z, f', f)$ on the right-hand side of the equation (34), determining the coefficients of this expansion by collecting the powers of $z$ and solving the resulting system of equations is a cumbersome procedure. Equations of the form (34) are often encountered in physics, and either their solutions can be determined numerically or their behavior is known qualitatively from experiments. Guided by this prior knowledge

about the nature of $f(z)$, we can eventually construct or guess a function $\phi_\gamma(z) = \phi(\gamma, z)$ that is a more favorable base for a power series than the independent variable $z$ itself. We have to cast the representation of the solution of equation (34) into the form

$$f(z) = f_0 + \sum_{n=1}^{\infty} C_n\big(f;\, \phi_\gamma, z_0\big)\big(\phi_\gamma(z) - \phi_\gamma(z_0)\big)^n,$$

and so we expand the solution $f(z)$ using the recursive formula (5). The code below gives an expansion of the first four terms.

```
Block[{f, z, z0, ϕ, y, h, g},
 Text@TraditionalForm@
   Row[{f[z], " = ",
     Row[
      Prepend[Table[c[f, ϕ, n, y] (ϕ[z] - ϕ[y])^n, {n, 3}],
             f[z0]] /. f''[y] → h[y, f'[y], f[y]] /. y → z0 /.
          h[z0, f'[z0], f[z0]] → h0 /. f'[z0] → g0 /.
        f[z0] → f0 /. z0 → z0 /. ϕ → ϕγ, "+"]
    }]
]
```

$$f(z) = f_0 + \frac{g_0\,(\phi_\gamma(z)-\phi_\gamma(z_0))}{\phi_\gamma'(z_0)} + \frac{(\phi_\gamma(z)-\phi_\gamma(z_0))^2\,(h_0\,\phi_\gamma'(z_0)-g_0\,\phi_\gamma''(z_0))}{2\,\phi_\gamma'(z_0)^3} + \frac{(\phi_\gamma(z)-\phi_\gamma(z_0))^3\,\big(f^{(3)}(z_0)\,\phi_\gamma'(z_0)^2-g_0\,\phi_\gamma^{(3)}(z_0)\,\phi_\gamma'(z_0)+3\,g_0\,\phi_\gamma''(z_0)^2-3\,h_0\,\phi_\gamma'(z_0)\,\phi_\gamma''(z_0)\big)}{6\,\phi_\gamma'(z_0)^5}$$

The free parameter $\gamma$ occurring in $\phi_\gamma$ will then be determined by the boundary conditions for $f(z)$. In the following, this kind of expansion with suitable basis functions will be applied to the solution of a problem of nonlinear heat transfer, and its convergence will be treated as far as relevant for this special case. For all the cases investigated in this article, we apply the recursive approach (7), since the structure of the chosen basis functions is relatively simple. For more sophisticated basis functions, the combinatorial formulas (12) and (14) have to be implemented in order to reduce CPU time. This may be done in future investigations.

## □ Heat Transfer in ZnO

As a canonical example, we study the partial differential equation of transient nonlinear heat transfer with temperature-dependent thermal conductivity $K(T)$. We demonstrate the application of Bürmann series to the practical problem of heat transfer in ZnO ceramics. The half space $z \geq 0$ is filled by this material, which has initial constant temperature $T_0$, and the temperature $T_1 \to T_0$ as $z \to \infty$. At $t = 0$ the surface temperature at $z = 0$ is instantaneously raised to a constant temperature $T_S$. Using the results of measurements of the thermal conductivity in ZnO [3], we can formulate the problem by writing

$$\frac{1}{\kappa_0}\,\frac{\partial T}{\partial t} = \frac{\partial}{\partial z}\left(K(T)\,\frac{\partial T}{\partial z}\right),$$
$$K(T) = \frac{1}{\frac{T}{T_1} - B}, \tag{35}$$
$$\kappa_0 = 5.38 \cdot 10^{-6}\left(\mathrm{m}^2 / \mathrm{s}\right), T_1 = 300\,(\mathrm{K}), B = 0.3133,$$
$$T(0, t) = T_S = 1200\,(\mathrm{K}), T(z, 0) = T_0 = 300\,(\mathrm{K}).$$

Using the transformation

$$T(z, t) / T_1 = \theta(z, t),$$

Kirchhoff's transformation

$$\Theta(z, t) = \ln(\theta(z, t) - B),$$

and Boltzmann's transformation

$$\eta = \frac{z}{2\sqrt{\kappa_0\, t}},$$

we find a nonlinear ordinary differential equation that has been extensively studied by Fujita [13–15], Lee [16, 17], and Crank [2]:

$$1.\ \frac{d^2\,\Theta}{d\eta^2} = -2\,\eta\,e^{\Theta}\,\frac{d\Theta}{d\eta},$$
$$2.\ \Theta(0) = \alpha = \ln\left(\frac{T_s}{T_1} - B\right), \Theta'(0) = -\sigma, \sigma > 0, \tag{36}$$
$$3.\ \Theta(\eta)_{\eta\to\infty} = \omega = \ln\left(\frac{T_0}{T_1} - B\right).$$

While the first boundary condition in (35) is regular (i.e. $T\,(0, t) = T_S$ and $\theta(0) = \alpha$), the second one is given in terms of the asymptotic expression $\lim_{z\to\infty} T(z, t) = T_1$. The equivalent value $\sigma$ of the derivative $(d\,\Theta / d\,\eta)_{\eta=0}$ has to be estimated, which is performed by calculating the time evolution of the total thermal energy of the semi-infinite half space [18]. The approximate value of this energy integral (in terms of an algebraic expression) is determined in appendix B, where we use the Bürmann series (10) to approximate the energy integral. The explicit expression is displayed in appendix B, equation (56), which describes the dependence of $\sigma$ on the parameters with a relative accuracy of 0.37%.

### ■ *Iterative Solution of Equation (36.1)*

To apply the methods developed in the preceding sections, we establish a convergent iteration scheme for equation 1 of (36). To this end, we transform this equation into an integral equation:

$$\Theta(\eta) = \alpha - \sigma \int_0^{\eta} d\xi \exp\left(-2 \int_0^{\xi} d\theta\, \theta\, e^{\Theta(\theta)}\right). \tag{37}$$

The solutions for 1 of equation (36) are strictly decreasing functions of $\eta$, which implies that for all positive values of $\eta$ we have

$$\Theta(\eta) \leq \alpha.$$

From this relation, we conclude that we can define a system of functions $\Psi_0, \Psi_1, \Psi_2, \ldots$ of the form

$$\begin{aligned} \Psi_0(\eta) &= \alpha - \sigma \int_0^{\eta} d\xi \exp\left(-2 \int_0^{\xi} d\theta\, \theta\, e^{\alpha}\right) = \alpha - \frac{\sigma}{2} \sqrt{\frac{\pi}{e^{\alpha}}}\ \mathrm{erf}\left(\sqrt{e^{\alpha}}\ \eta\right), \\ \Psi_k(\eta) &= \alpha - \sigma \int_0^{\eta} d\xi \exp\left(-2 \int_0^{\xi} d\theta\, \theta\, e^{\Psi_{k-1}(\theta)}\right). \end{aligned} \tag{38}$$

Taking the first term of the expansion (32) of the error function, we have

$$\begin{aligned} &\Theta(\eta) \leq \Psi_k(\eta) \leq \Psi_{k-1}(\eta) \leq \ldots \leq \alpha - \frac{\sigma}{2} \sqrt{\frac{\pi}{e^{\alpha}}}\ \mathrm{erf}\left(\sqrt{e^{\alpha}}\ \eta\right) \leq \varphi(\eta), \\ &\varphi(\eta) = \alpha - \frac{\sigma}{2} \sqrt{\frac{\pi}{e^{\alpha}}}\ \sqrt{1 - e^{-e^{\alpha}\eta^2}}\ . \end{aligned} \tag{39}$$

## □ Enhancing the Convergence of the Bürmann Series

Since the system (39) converges toward the solution $\Theta(\eta)$, a possible choice for a basis function would be $\varphi(\eta)$. Instead of $\varphi(\eta)$, we prefer to introduce a less complicated function $\phi_\gamma(\eta)$ that simplifies the calculations. According to (39), this function has to fulfill the following conditions:

$$\Theta(\eta) \leq \phi_\gamma(\eta),\ 0 \leq \eta < \eta_\omega,\ \eta_\omega \longrightarrow \infty,$$

$$\phi_\gamma(0) = \Theta(0) = \alpha.$$

The function $\phi_\gamma(\eta)$ can be constructed in such a way as to guarantee that all essential boundary conditions are fulfilled by $\Theta(\eta)$:

$$\phi_\gamma(\eta) = \alpha - \frac{\sigma}{\gamma}\, \psi_\gamma(\eta),$$

$$\phi_\gamma(0) = \alpha,\ \phi_\gamma{}'(0) = -\sigma.$$

The parameter $\gamma$ is as yet undetermined. A useful basis for an expansion is obtained by taking

$$\psi_\gamma(\eta) = 1 - e^{-\gamma\,\eta}, \tag{40}$$

which will lead, according to (6) and (7), to the Bürmann series calculated as follows.

```
Module[{d},
 kend = 5; (* maximum order of expansion *)
 d[1][z_] = Θ'[z]/φ'[z];
 ϕ[γ_, σ_, z_] = α - σ/γ (1 - Exp[-γ z]);
 Do[
  d[k][z_] =
   Simplify[
    (1/(k φ'[z]) D[d[k - 1][x], x] /. Θ''[x] → -2 x E^Θ[x] Θ'[x] /.
       x → z)], {k, 2, kend}
 ];
 Do[
  a[k] = Limit[D[ϕ[γ, σ, z], {z, k}], z → 0];
  c1[k] =
   Simplify[(d[k][z] /. z → 0) /.
     Join[{Θ[0] → α, Θ'[0] → -σ},
      Table[φ^(i)[0] → a[i], {i, 1, kend}]]],
  {k, 1, kend}
 ];
 Θ[1, α_, γ_, σ_, η_] := ϕ[γ, σ, η];
 Θ[k_, α_, γ_, σ_, η_] :=
  Θ[k - 1, α, γ, σ, η] + c1[k] (ϕ[γ, σ, η] - α)^k;
]
```

```
Θ[5, α, γ, σ, η]
```

$$\alpha - \frac{(1-e^{-\gamma\eta})\,\sigma}{\gamma} - \frac{(1-e^{-\gamma\eta})^2\,\sigma}{2\,\gamma} - \frac{(1-e^{-\gamma\eta})^3\,\left(-e^{\alpha}+\gamma^2\right)\sigma}{3\,\gamma^3} + \frac{(1-e^{-\gamma\eta})^4\,\left(-3\,\gamma^3+e^{\alpha}\,(6\,\gamma-2\,\sigma)\right)\sigma}{12\,\gamma^4} - \frac{(1-e^{-\gamma\eta})^5\,\sigma\,\left(6\,e^{2\,\alpha}+12\,\gamma^4+e^{\alpha}\left(-35\,\gamma^2+20\,\gamma\,\sigma-3\,\sigma^2\right)\right)}{60\,\gamma^5}$$

Expressed in terms of $\phi_\gamma(\eta)$, this is

$$\Theta(\eta) = \phi_\gamma(\eta) - \frac{\gamma}{2\,\sigma}\left(\phi_\gamma(\eta) - \alpha\right)^2 + \frac{1}{3}\left(\frac{\gamma^2 - e^\alpha}{\sigma^2}\right)\left(\phi_\gamma(\eta) - \alpha\right)^3 + \left(2\,e^\alpha(3\,\gamma - \sigma) - 3\,\gamma^3\right)/\left(12\,\sigma^3\right)\left(\phi_\gamma(\eta) - \alpha\right)^4 + \frac{1}{60\,\sigma^4}\left(6\,e^{2\,\alpha} + 12\,\gamma^4 + e^\alpha\left(20\,\gamma\sigma - 3\,\sigma^2 - 35\,\gamma^2\right)\right)\left(\phi_\gamma(\eta) - \alpha\right)^5 + \ldots\,. \tag{41}$$

## Convergence of the Enhanced Bürmann Series

If we consider $\gamma$ in (40) and (41) as a free parameter, we have to investigate how its choice influences the convergence of the corresponding Bürmann series. We write, using (37) and omitting the index $\gamma$ for $\psi$,

$$\Phi_1(\eta) = \alpha - \frac{\sigma}{\gamma}\int_0^{\gamma\eta} d\zeta\,\exp\left(-2\left(\frac{\beta}{\gamma}\right)^2\int_0^\zeta d\theta\,\theta\,e^{-\lambda\,\psi(\theta)}\right)$$
$$= \alpha - \lambda\int_0^{\psi(\eta)}\frac{d\psi}{1-\psi}\exp\left(-2\left(\frac{\beta}{\gamma}\right)^2\int_0^\psi d\psi^*\,\ln\left(\frac{1}{1-\psi^*}\right)\frac{1}{1-\psi^*}\,e^{-\lambda\,\psi^*}\right),$$
$$\lambda = \frac{\sigma}{\gamma},\ \beta = \sqrt{e^\alpha}\,.$$

All integrands are representable by uniformly convergent series expansions in powers of $\psi < 1$, and thus the argument is similar to the one used in proving the convergence of the system (39). Furthermore, we observe that the expansion of the exponential function $\exp\left(-2\,(\beta/\gamma)^2\int d\theta\ldots\right)$ converges more quickly for higher values of $\gamma$. Thus, as long as $\gamma$ can be chosen to guarantee the condition

$$\Theta(\eta) \leq \alpha - \frac{\sigma}{\gamma}\,\psi_\gamma(\eta)$$

for some value $\gamma$, an iterative system of functions can be constructed that converges to the solution of equation 1 of (36). This fact can be exploited by determining $\gamma$ in such a way as to assure that the approximation

$$\Theta_k(\eta) = \alpha - \sigma\sum_{n=1}^{k} C_n\left(-\frac{\sigma}{\gamma}\,\psi_\gamma(\eta)\right)^n$$

assumes the correct value $\omega$ at infinity. So we have

$$\lim_{\eta\to\infty}\Theta_k(\eta) = \lim_{\eta\to\infty}\left(\alpha - \sigma\sum_{n=1}^{k} C_n\left(-\frac{\sigma}{\gamma}\,\psi_\gamma(\eta)\right)^n\right) = \omega, \tag{42}$$

resulting in an algebraic equation of the $k^{\text{th}}$ order in $\gamma$ with roots $\gamma_k$,

$$\alpha - \sigma \sum_{n=1}^{k} C_n \left(-\frac{\sigma}{\gamma}\right)^n = \omega, \tag{43}$$

where $\gamma_k$ satisfies the condition

$$\Theta(\eta) \leq \alpha - \frac{\sigma}{\gamma_k} \psi_{\gamma_k}(\eta), 0 \leq \eta < \infty. \tag{44}$$

## □ The Cubic Approximation

If there is a solution $\gamma_k$ to equation (43) simultaneously fulfilling condition (44), we get an approximation $\Theta_k$ that converges to the correct value $\omega$ of $\Theta$ as $\eta \to \infty$. For the third-order approximation $\Theta_3$, we get from (41) and (43) the cubic equation

$$\gamma^3 - \frac{11\,\sigma}{6\,(\alpha - \omega)} \gamma^2 + \frac{\sigma\, e^{\alpha}}{3\,(\alpha - \omega)} = 0. \tag{45}$$

The relevant real solution to this equation is

$$\gamma_3 = \frac{11}{18} \frac{\sigma}{(\alpha - \omega)} \left(1 + 2 \sin\left(\frac{\vartheta}{3}\right)\right), \tag{46}$$

with

$$\begin{aligned} \vartheta &= 2\,\pi + \sin^{-1}\left(\frac{\sqrt{27}}{2}\, \mathfrak{w}\right), \\ \mathfrak{w} &= \sqrt{\frac{4}{27} \left(\frac{972}{1331} \left(\frac{\alpha - \omega}{\sigma}\right)^2 e^{\alpha} - 1\right)}. \end{aligned} \tag{47}$$

We display the explicit expression for the third-order approximation below:

$$\begin{aligned} &T(z, t) \approx \\ &\quad (T_S - B\,T_1) \exp\left(-\lambda_3\, \psi_{\gamma_3}\left(1 + \frac{1}{2} \psi_{\gamma_3} + \frac{1}{3}\left(1 - \frac{T_S - B\,T_1}{{\gamma_3}^2\, T_1}\right) {\psi_{\gamma_3}}^2\right)\right) + B\,T_1, \\ &\psi_{\gamma_3}(z, t) = 1 - e^{-\frac{\gamma_3 z}{\sqrt{4 \kappa_0 t}}}, \lambda_3 = \frac{\sigma}{\gamma_3} = \frac{18}{11} \ln\left(\frac{T_S - B\,T_1}{T_0 - B\,T_1}\right) \Big/ \left(1 + 2 \sin\left(\frac{\vartheta}{3}\right)\right). \end{aligned} \tag{48}$$

In the next two plots, we show a comparison between the approximation (48) and the exact solution found by applying `NDSolve` to 1 of (36). The value for $\sigma$ is calculated by using the approximation $\sigma_0$ given by equation 2 of (57). Note that (48) can also be inverted exactly by using common algebraic and transcendent functions. The corresponding procedure is listed below for the same parameters as given in the previous section.

This defines the cubic approximation according to (45)–(48).

```
Tcubic[z_, t_,
  κ0_, (* thermal diffusivity [m^2/sec] (35) *)
  TS_, (* Surface temperature [Kelvin]*)
  T0_, (* Initial temperature [Kelvin] of the half
   space *)
  T1_, (* Temperature [Kelvin] of the half space
   at infinity *)
  b_, (* parameter B in (35) *)
  σ_ (* parameter as def. in (36.2) calculated by
   equ. (57.2) *)
 ] := Module[{α, ω, ѡ, ϑ, γ3, λ3},
  α = Log[TS / T1 - b]; (* definition of α according
   equ. 36.2 *)

  ω = Log[T0 / T1 - b]; (* parameter as def. in (36.3) *)
  ѡ = Sqrt[4/27] (972/1331 ((α - ω)/σ)^2 ⅇ^α - 1);
  ϑ = 2 π + ArcSin[Sqrt[27]/2 ѡ];
  γ3 = 11/9 σ/(α - ω) (Sin[ϑ/3] + 1/2);
  λ3 = σ/γ3;
  (TS - b T1)
    Exp[ -λ3 (1 - ⅇ^(-(γ3 z)/Sqrt[4 κ0 t]))
      (1 + 1/2 (1 - ⅇ^(-(γ3 z)/Sqrt[4 κ0 t])) + 1/3 (1 - ⅇ^α/γ3^2) (1 - ⅇ^(-(γ3 z)/Sqrt[4 κ0 t]))^2) ] + b T1
 ]
```

For the numerically exact solution, calculated by using `NDSolve`, we impose the condition $\Theta(\infty) = 0$ (i.e. $T(\infty, t) = T_0$). This condition is equivalent to selecting a slope $\sigma = \Theta'(0)$ at the surface $z = 0$. Bürmann's theorem is used a second time to find an approximation $\sigma_0 \approx 2.23136$ for the slope $\sigma$ by calculating the Bürmann expansion (56) of the energy integral (see appendix B).

```
Texact[z_, t_, κ0_, TS_, T1_, b_, σ_] :=
 (* solution in real space in SI-units *)
 T1 (Exp[Θ[z/(2 Sqrt[t κ0])]] + b) /.
  NDSolve[{Θ''[η] == -2 η Θ'[η] Exp[Θ[η]],
    Θ[0] == Log[TS / T1 - b], Θ'[0] == -σ}, Θ, {η, 0, 10}
   (* numerical solution of transformed equation,
   slope σ at η=0 replacing T(∞,t)=T0*)]
```

The plot shows the numerically exact temperature profiles (the colored curves) and the exact solution's third-order approximation according to equation (48) (the dotted lines), in the range $z = 0 - 0.25$ m at $t = 10, 30, 100, 300, 1000, 3000, 10\,000$ s.

```
Module[{κ0 = .38 × 10^-6 , TS = 1200, T0 = 300 , T1 = 300,
   b = 0.3133, σ = 2.23136
   (* approximation σ=σ0( ≈ 2.23136) equ. (57.2) *),
   tarray, m},
  tarray = {10, 30, 100, 300, 1000, 3000, 10 000};
  (* time steps for evaluation *)
  m = Map[Row[{Style["t", Italic], " = ", #, " s"}] &,
    tarray];
  Show[
   Plot[
    MapThread[Tooltip, {
        Map[Tcubic[z, #, κ0, TS, T0, T1, b, σ] &, tarray],
        Map[Row[{Style["t", Italic], " = ", #, " s"}] &,
          tarray] // Evaluate
       }] // Evaluate,
    {z, 0, 0.25},
    PlotStyle → {{Black, Thickness[0.005], Dotted}},
    PlotRange → All
   ],
   Plot[
    MapThread[Tooltip, {
        Map[Texact[z, #, κ0, TS, T1, b, σ] &, tarray],
        Map[Row[{Style["t", Italic], " = ", #, " s"}] &,
          tarray] // Evaluate
       }] // Evaluate,
    {z, 0, 0.25}, PlotRange → All
   ],
   Frame → True,
   FrameLabel → {Row[{z , " [m]"}], "temperature [Kelvin]"},
   PlotLabel ->
    "Compare exact solution and cubic approximation",
   ImageSize → 400
  ]
 ]
```




Additionally the relative error of the third-order approximation (48) is displayed for the same profiles as shown before.

```
Module[{κ0 = .38 × 10^-6 , TS = 1200, T0 = 300 , T1 = 300,
   b = 0.3133, σ = 2.23136, tarray, m},
  tarray = {10, 30, 100, 300, 1000, 3000, 10 000};
  (* time steps for evaluation *)
  m = Map[Row[{Style["t", Italic], " = ", #, " s"}] &,
    tarray];
  Plot[
   MapThread[Tooltip,
     {Map[
       (Texact[z, #, κ0, TS, T1, b, σ] -
           Tcubic[z, #, κ0, TS, T0, T1, b, σ]) /
         Texact[z, #, κ0, TS, T1, b, σ] &, tarray],
      Map[Text@Row[{Style["t", Italic], " = ", #, " s"}] &,
       tarray]}] // Evaluate, {z, 0, 0.25},
PlotRange → All, Frame → True,
   FrameLabel -> {Row[{Style["z", Italic], " [m]"}],
     "relative error"},
   PlotLabel -> "Relative error of the cubic approximation",
   ImageSize → 400]
]
```




## Summary and Conclusion

The goal of this work is to give a comprehensive presentation of Bürmann's theorem and its application to linear and nonlinear DEs and PDEs of heat transfer, using single-valued and multivalued basis functions.

To this end, a reformulation of the formulas of the expansion coefficients of Bürmann series, based on a combinatorial viewpoint, is developed. As a result of this reformulation, an algorithm is presented, which accelerates the calculation of expansion coefficients, compared to standard methods. Using this approach, the expansion of transcendental functions in powers of their derivative is applied to the error integral, to the solution of nonlinear differential equations, and to the evaluation of the heat integral. By combining these methods, it is possible to show that the approximate solution of nonlinear problems of heat transfer can be given in terms of Bürmann expansions. Finally, it is shown that the introduction of an additional parameter in the basis function can significantly enhance the convergence of a Bürmann series. The value of this parameter can be found by solving algebraic equations that result from the boundary conditions of the problems.

# Appendix A

## Expansion Coefficients of $f(z)^{-m}$

The coefficients $\mathrm{R}_k(f, m)$, defined by

$$\frac{1}{f(z)^m} = \frac{1}{f(z_0)^m} \sum_{k=0}^{\infty} \mathrm{R}_k(f, m)\,(z - z_0)^k,$$

result from elementary considerations. We write

$$1.\ \frac{1}{f(z)^m} =$$
$$\frac{1}{f(z_0)^m}\left(\frac{1}{1+S}\right)^m = \frac{1}{f(z_0)^m}\sum_{n=0}^{\infty}(-1)^n\,(m+n-1)!\,/\,((m-1)!\,n!)\,S^n, \tag{49}$$
$$2.\ S = \sum_{k=1}^{\infty}\frac{f^{(k)}(z_0)}{k!\,f(z_0)}\,\zeta^k,\ \zeta = (z - z_0).$$

The $n^{\text{th}}$ power of $S$ is given by

$$1.\ S^n =$$
$$n!\sum_{\text{all sets of } q(n)}\frac{1}{q_{u_1}(n)!}\left(\frac{f^{(u_1)}(z_0)}{u_1!\,f(z_0)}\right)^{q_{u_1}(n)}\frac{1}{q(n)_{u_i}!}\left(\frac{f^{(u_2)}(z_0)}{u_2!\,f(z_0)}\right)^{q_{u_2}(n)}\cdots\frac{1}{q_{u_n}(n)!}$$
$$\left(\frac{f^{(u_n)}(z_0)}{u_n!\,f(z_0)}\right)^{q_{u_n}(n)}\zeta^{u_1\,q_{u_1}(n)+u_2\,q_{u_2}(n)+\ldots\mathrm{u}_n\,q_{u_n}(n)}, \tag{50}$$
$$2.\ \sum_{i=1}^{n} q_{u_i}(n) =$$
$$n.$$

Rearranging equation 1 of (50) in increasing powers $\zeta^{k(n)}$, we have two conditions

$$1.\ \sum_{i=1}^{n} q_{u_i}(n) = n,$$
$$2.\ \sum_{i=1}^{n} u_i\,q_{u_i}(n) = k(n). \tag{51}$$

Thus, collecting all $k(n)^{\text{th}}$ powers of $\zeta$ over all the contributions arising from all $S^n$ with $1 \leq n \leq k$ is equivalent to imposing one single condition, replacing the conditions 1 and 2 of (51):

$$\sum_{s=1}^{k} s\, p_s^k = k.$$

Since $k \geq \sum_{s=1}^{k} p_s^k$, we define $p_0^k \geq 0$:

$$k - \sum_{s=1}^{k} p_s^k = p_0^k. \tag{52}$$

Using equations 1 of (49), 1 of (50), and (52) we finally arrive at the expansion

$$\frac{1}{f(z)^m} =$$

$$\frac{1}{f(z_0)^m} \sum_{k=0}^{\infty} \sum_{\substack{\text{all partitions } p \text{ of } k, \\ \text{freq}(p)=(p_1,\ldots,p_k)}} (-1)^{(k-p_0)} \frac{(m+k-p_0-1)!}{(m-1)!} \prod_{s=1}^{k} \frac{1}{p_s!} \left( \frac{f^{(s)}(z_0)}{s!\, f(z_0)} \right)^{p_s} (z-z_0)^k.$$

A similar result, displayed in (21), is obtained for $m = \mu / \nu$.

## Appendix B

### Bürmann Series (10) for the Heat Integral

We define the temperature difference $\Delta T$ and the differential equation it obeys:

$$\begin{aligned} &1.\ \Delta T = T - T_0, \\ &2.\ \frac{1}{\kappa_0} \frac{\partial \Delta T}{\partial t} = \frac{\partial}{\partial z} \left( K(T) \frac{\partial \Delta T}{\partial z} \right). \end{aligned} \tag{53}$$

By integrating equation 2 of (53) over $z$, we find

$$\frac{1}{\kappa_0} \frac{d}{dt} \int_0^{\infty} \Delta T\, dz = \frac{1}{\kappa_0} \frac{d\Xi(t)}{dt} = -K(T_S) \left( \frac{\partial T}{\partial z} \right)_{z=0}. \tag{54}$$

Performing the transformations of Kirchhoff and Boltzmann, we arrive at the solution for $\Xi(t)$:

$$\frac{d\Xi(t)}{dt} = \frac{\sigma\, T_1 \sqrt{\kappa_0}}{2} \frac{1}{\sqrt{t}} \longrightarrow \Xi(t) = \sigma\, T_1 \sqrt{\kappa_0\, t}\,. \tag{55}$$

On the other hand, using the definition given in (54), we expand $\Xi(t)$ in a Bürmann series in powers of $\Delta T$ according to (10). This leads to

$$\begin{aligned}\Xi(t) &= \int_0^\infty \Delta T\, dz \\ &= 2\sqrt{\kappa_0\, t}\int_0^\infty \Delta T\, d\eta \\ &= 2\sqrt{\kappa_0\, t}\sum_{k=1}^{\infty}(-1)^k\, C_k(\Xi;\, \Delta T, 0)\, {\Delta T_0}^k,\, \Delta T_0 = T_S - T_0,\, \Delta T_\infty = 0.\end{aligned} \tag{56}$$

Combining (55) and (56) up to order three gives

$$\begin{aligned}&-\frac{\Delta T_0}{\Delta T_0'}\,\Delta T_0 + \frac{1}{2}\,\frac{{\Delta T_0'}^2 - \Delta T_0\,\Delta T_0''}{{\Delta T_0'}^3}\,{\Delta T_0}^2 - \\ &\quad \frac{1}{6}\,\frac{1}{{\Delta T_0'}^5}\left(3\,\Delta T_0\,{\Delta T_0''}^2 - 2\,{\Delta T_0'}^2\,\Delta T_0'' - \Delta T_0\,\Delta T_0'\,{\Delta T_0}^{(3)}\right){\Delta T_0}^3 = \frac{\sigma}{2}\,T_1.\end{aligned}$$

According to the transformations in (36), we have

$$\begin{aligned}\Theta'' &= -2\,\eta\,\Theta'\,e^{\Theta}, \\ \Delta T_0' &= \left(\frac{dT}{d\eta}\right)_{\eta=0} = T_1\,e^{\Theta_0}\,\Theta_0' = -T_1\,e^{\alpha}\,\sigma, \\ \Delta T_0'' &= \left(\frac{d^2\,T}{d\eta^2}\right)_{\eta=0} = T_1\,e^{\Theta_0}\,{\Theta_0'}^2 + T_1\,e^{\Theta_0}\,\Theta_0'' = T_1\,e^{\alpha}\,\sigma^2, \\ {\Delta T_0}^{(3)} &= \left(\frac{d^3\,T}{d\eta^3}\right)_{\eta=0} = -T_1\,e^{\alpha}\,\sigma^3 + 2\,T_1\,e^{2\,\alpha}\,\sigma,\end{aligned}$$

which leads, after some manipulations, to a quartic algebraic equation for $\sigma$ and its solution $\sigma_0$, given by

$$\begin{aligned}&1.\,\sigma^4 + A\sigma^2 - B = 0, \\ &2.\,\sigma_0 = \frac{1}{\sqrt{2}}\sqrt{\sqrt{A^2 + 4\,B} - A}\,, \\ &3.\,A = \left((T_S - T_0)^2\left(T_0(5\,T_S - BT_1) + 7\,BT_1\,T_S - 6\,T_S^2 - 2\,T_0^2 - 3\,B^2\,T_1^2\right)\right)\Big/ \\ &\quad \left(3\,T_1(T_S - BT_1)^3\right), \\ &4.\,B = \frac{2\,(T_S - T_0)^4}{3\,T_1^2(T_S - BT_1)^2}.\end{aligned} \tag{57}$$

An approximation of (56) up to order six would lead to a sextic equation that is reducible to a cubic equation, and hence to an algebraic expression for $\sigma$. The approximation obtained from the first equation of (57) for $\sigma$ is $\sigma_0 = 2.23136$, which shows a maximum relative error of 0.37% compared to the exact value $\sigma_{\text{exact}} = 2.22313$ found by numerical methods (i.e. `NDSolve`) using the parameters listed in (35).

## ■ Acknowledgments

The authors wish to thank Prof. Dieter Messner, Lienz, and Dr. Hans Riedler, Graz, for their encouragement and helpful advice. Special thanks go to the editors and reviewers of *The Mathematica Journal* for their constructive suggestions and support.

## ■ References

[1] H. S. Carslaw and J. C. Jaeger, *Conduction of Heat in Solids*, 2nd ed., Oxford: Clarendon Press, 1959 pp. 482–484.

[2] J. Crank, *The Mathematics of Diffusion*, 2nd ed., Oxford: Clarendon Press, 1975 pp. 107–110, 119–121.

[3] T. Olorunyolemi, A. Birnboim, Y. Carmel, O. C. Wilson Jr., I. K. Loyd, S. Smith, and R. Campbell, "Thermal Conductivity of Zinc Oxide: From Green to Sintered State," *Journal of the American Ceramic Society*, **85**(5), 2002 pp.1249–53. doi:10.1111/j.1151-2916.2002.tb00253.x.

[4] C. Wagner, "Diffusion of Lead Chloride Dissolved in Solid Silver Chloride," *Journal of Chemical Physics*, **18**, 1950 pp. 1227–1230. doi:10.1063/1.1747915.

[5] E. T. Whittaker and G. N. Watson, *A Course of Modern Analysis*, 4th ed., Cambridge: Cambridge University Press, 1927 pp. 128–132.

[6] E. W. Weisstein. "Bürmann's Theorem" from Wolfram *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/BuermannsTheorem.html.

[7] Wikipedia. "Lagrange Inversion Theorem." (Oct 17, 2014) en.wikipedia.org/wiki/Lagrange_inversion_theorem.

[8] H. Stenlund, "Inversion Formula." arxiv.org/abs/1008.0183.

[9] E. W. Weisstein, "Series Reversion" from *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/SeriesReversion.html.

[10] L. Comtet, "*Advanced Combinatorics: The Art of Finite and Infinite Expansions*" (J. W. Nienhuys, trans.), D. Reidel: Dordrecht, 1974.

[11] E. W. Weisstein, "Faà di Bruno's Formula" from *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/FaadiBrunosFormula.html.

[12] P. M. Morse and H. Feshbach, *Methods of Theoretical Physics*, Vol. I, New York: McGraw-Hill, 1953 pp. 411–413.

[13] H. Fujita, "The Exact Pattern of a Concentration-Dependent Diffusion in a Semi-infinite Medium, Part I," *Textile Research Journal*, **22**(11), 1952 pp. 757–760. doi:10.1177/004051755202201106.

[14] H. Fujita, "The Exact Pattern of a Concentration-Dependent Diffusion in a Semi-infinite Medium, Part II," *Textile Research Journal*, **22**(12), 1952 pp. 823–827. doi:10.1177/004051755202201209.

[15] H. Fujita, "The Exact Pattern of a Concentration-Dependent Diffusion in a Semi-infinite Medium, Part III," *Textile Research Journal*, **24**(3), 1954 pp. 234–240. doi:10.1177/004051755402400304.

[16] C. F. Lee "On the Solution of Some Diffusion Equations with Concentration-Dependent Diffusion Coefficients—I," *Journal of the Institute of Mathematics and Its Applications* (now *IMA Journal of Applied Mathematics*), **8**(2), 1971 pp. 251–259. doi:10.1093/imamat/8.2.251.

[17] C. F. Lee "On the Solution of Some Diffusion Equations with Concentration-Dependent Diffusion Coefficients—II," *Journal of the Institute of Mathematics and Its Applications* (now *IMA Journal of Applied Mathematics*), **10**(2), 1972 pp. 129–133. doi:10.1093/imamat/10.2.129.

[18] T. R. Goodman, "Application of Integral Methods for Transient Nonlinear Heat Transfer," in *Advances in Heat Transfer*, Vol. I (T. F. Irvine, Jr. and J. P. Hartnett, eds.), New York: Academic Press, 1964 pp. 51–122.

H. M. Schöpf and P. H. Supancic, "On Bürmann's Theorem and Its Application to Problems of Linear and Nonlinear Heat Transfer and Diffusion," *The Mathematica Journal*, 2014. dx.doi.org/doi:10.3888/tmj.16-11.

## About the Authors

Harald Markus Schöpf was born in Lienz, Austria on May 5, 1965. He studied solid-state physics and technical physics in Graz. After two years as a research assistant at the Graz University of Technology, Austria, he moved to Siemens Matsushita (now EPCOS) in Deutschlandsberg, Austria. Now he works as a teacher in Lienz.

Peter Hans Supancic was born in Graz, Austria on September 7, 1967. He studied theoretical physics in Graz and graduated in materials science at the University of Leoben, Austria. After finishing the habilitation on functional ceramics, he became ao. Prof. at the University of Leoben.

**Harald Markus Schöpf**
*Schillerstrasse 4*
*A-9900 Lienz*
*harald.schoepf@gmx.net*

**Peter Hans Supancic**
Institut für Struktur- und Funktionskeramik/Montanuniversitaet Leoben
*Peter Tunner Strasse 5*
*A-8700 Leoben*
*peter.supancic@unileoben.ac.at*